
Data

Data come from the Consumer Expenditure Survey (CEX) from 2004-2019, which
provides detailed information on the expenditures, income, and demographics of a
large nationally representative sample of households in the United States. I use data
from the interview survey, which is designed to obtain data on larger and reoccurring
expenditures that respondents can recall for several months. Each household is
interviewed each quarter for four consecutive quarters. I start with 2004 because
this is the first year that missing income data was imputed. I limit the sample to
households with resident children under age 18, where at least one parent in the
household is over the age of 24 and neither parent is over 65. In all, I analyze
expenditures for children from 40,373 families in 110,034 household-quarters. These
microdata are publicly available from the U.S. Bureau of Labor Statistics at https:

//www.bls.gov/cex/pumd.htm

Measures

Parental Investments: Consistent with previous work, I define parental investments
as the total sum of expenditures on childcare, schooling, and enrichment activities,
and I divide this by the number children in the household under age 18 to generate a
per child expenditure measure (for further rationale, see Hastings and Schneider 2021;
Kornrich 2016; Kornrich and Furstenberg 2013; Schneider, Hastings, and LaBriola
2018). The data are adjusted to 2021 real dollars using the CPI-U-RS and multiplied
by 4 to represent annual, rather than quarterly, spending.

Race/Ethnicity: Data on racial identity and Hispanic status are collected on the
household head and partner (if present), but not for the children. Race/ethnicity is
determined by the self-reported race/ethnicity of the mother in each household, unless
the mother was not present, in which case the father’s was used. I coded race/ethnicity

with categories for (1) white, Non-Hispanic (60% of sample); (2) Black, non-Hispanic
(11%); (3) Hispanic (21%) and (4) other race, non-Hispanic (8%) respondents. Given
the wide range of actual racial and ethnic identities captured in the other category, I
do not include this group in my analysis.

Income: Family income includes all sources of income and public assistance. Taxes are
estimated by the CEX using TAXSIM, which creates an after-tax/transfer measure
which is used here. Missing income data are imputed by CEX, and incomes are
adjusted to 2021 real dollars using the CPI-U-RS.

Family Characteristics: Family characteristic controls are family structure (married,
cohabiting, single), number of adults in the household, number of children in each age
category (0-5, 6-11, 12-17), highest education of household head or partner (no high
school diploma, high school diploma but no Bachelor’s degree, Bachelor’s degree or
above), oldest age and age-squared of household head or partner, and whether or not
a grandparent is in the home.

Methods

I run linear regression models of parental investments, employing the sampling weights
provided by the CEX and adjusting the standard errors for clustering by households
(since households can appear in up to four quarters of the data).

The first model includes only the race/ethnicity variable, month fixed effects (to
adjust for seasonality in parental investments) and year fixed effects (to adjust for
overall changes in spending over time). Formally, this model can be written as:

Parental Investments = β0 + β1Black + β2Hispanic+ µmonth + µyear

where β1 represents the Black-White gap and β2 represents the Hispanic-White gap.
This shows the real differences in parental investments by race and ethnicity. The
second model adds controls for family characteristics (but not income). The third model
adds a control for income (but not family characteristics). The final model controls
for both family income and characteristics. The models are shown in Table 1. The
visualization in the paper is created by taking the results of these models and predicting
the average level of spending by race/ethnicity, holding all of the control variables at
their mean values. The means are shown with 83.4% confidence intervals, which is
selected because this roughly corresponds to the level where non-overlapping errors bars
indicates a statistically significant difference at p < .05. For more explanation of the
error bars, see: https://chris-said.io/2014/12/01/independent-t-tests-and

-the-83-confidence-interval-a-useful-trick-for-eyeballing-your-data/.

Results

Table 1: Regression Models of Parental Investments

(1) (2) (3) (4)

Actual

Controlling
for family

characteristics
Controlling
for income

Controlling for
income & family
characteristics

White ref. ref. ref. ref.

Black −1021.4∗∗∗ −446.5∗∗∗ −252.5∗∗∗ −175.5∗∗

(57.9) (61.1) (60.3) (60.0)
Hispanic −1177.8∗∗∗ −266.2∗∗∗ −463.6∗∗∗ −31.1

(41.2) (41.2) (40.2) (40.3)
Income (in thousands) 17.6∗∗∗ 16.5∗∗∗

(0.62) (0.71)
Num. of adults in house −186.9∗∗∗ −341.2∗∗∗

(35.2) (36.7)
Num. of children 0-5 87.9∗∗∗ 75.7∗∗

(25.0) (23.6)
Num. of children 6-11 −421.9∗∗∗ −450.1∗∗∗

(24.1) (23.3)
Num. of children 12-17 −629.7∗∗∗ −664.2∗∗∗

(26.4) (25.8)
Married ref. ref.

Cohabiting −448.0∗∗∗ −198.5∗∗∗

(48.3) (45.6)
Single −314.7∗∗∗ 177.1∗∗∗

(55.0) (50.7)
Parental education (highest) 500.2∗∗∗ 225.7∗∗∗

(13.7) (13.6)
Grandparent present −19.3 −120.8

(85.9) (84.7)
Age 126.7∗∗∗ 49.8∗

(24.0) (22.8)
Age × Age −1.13∗∗∗ −0.44

(0.30) (0.28)
Constant 1754.6∗∗∗ −7347.1∗∗∗ −128.9 −3056.2∗∗∗

(80.0) (507.2) (97.9) (483.9)
Month Fixed Effects Y es Y es Y es Y es
Year Fixed Effects Y es Y es Y es Y es

Observations 110034 110034 110034 110034

Clustered standard errors in parentheses
∗ p < .05, ∗∗ p < .01, ∗∗∗ p < .001

Code

Analyses were conducted using Stata 17. User-written packages used are estout (Jann
2007) for the table, coefplot (Jann 2014) for the figure, and schemepack (Naqvi
2021) for graph scheme white tableau for the figure. The code to run the analysis
and create the visualization is given below:

cd /Users/oph/Dropbox/projects/parenting_race

use if year >= 2004 using /Users/oph/Dropbox/data/cex/cleaned_2020/CEX_2003_2019_cleaned_quarterly.dta, clear

replace total_edu = 4*total_edu // convert to annual dollars

drop if missing(inca1000)

lab var inca1000 "Income (in thousands)" // after taxes and transfers
label var total_edu "Total parental investments"
label var max_educ "Parental education (highest)"
label var agecat0_5 "Num. of children 0-5"
label var agecat6_11 "Num. of children 6-11"
label var agecat12_17 "Num. of children 12-17"
lab var adults "Num. of adults in house"
lab var cu_multigen "Grandparent present"

rename altrace raceeth // raceeth coding based on mothers race, unless mother not present

drop if raceeth == 4

global family adults agecat0_5 agecat6_11 agecat12_17 i.famstructure max_educ cu_multigen c.max_age##c.max_age

eststo M1: reg total_edu i.raceeth i.qintrvmo i.qintrvyr [pw = finlwt21], vce(cluster cu_id)

eststo M2: reg total_edu i.raceeth $family i.qintrvmo i.qintrvyr [pw = finlwt21], vce(cluster cu_id)

eststo M3: reg total_edu i.raceeth inca1000 i.qintrvmo i.qintrvyr [pw = finlwt21], vce(cluster cu_id)

eststo M4: reg total_edu i.raceeth inca1000 $family i.qintrvmo i.qintrvyr [pw = finlwt21], vce(cluster cu_id)

forvalues i = 1/4 {
est restore M`i'
margins raceeth, atmeans post
eststo P`i'

}

#delimit ;
coefplot

(P1,keep(1.raceeth)) ///
(P1,keep(2.raceeth)) ///
(P1,keep(3.raceeth)) ///
, bylabel(" " "Actual" " ") subtitle(, size(small)) legend(off)
||
(P2,keep(1.raceeth)) ///
(P2,keep(2.raceeth)) ///
(P2,keep(3.raceeth)) ///
, bylabel("Controlling" "for family" "characteristics") legend(off)
||
(P3,keep(1.raceeth)) ///
(P3,keep(2.raceeth)) ///
(P3,keep(3.raceeth)) ///
, bylabel("Controlling" "for" "income") legend(off)
||
(P4,keep(1.raceeth)) ///
(P4,keep(2.raceeth)) ///
(P4,keep(3.raceeth)) ///
, bylabel("Controlling for" "family characteristics" "and income") legend(off)
||
,
legend(off) msize(vlarge) nooffsets vertical recast(dropline)
ciopts(recast(rcap) lw(medthick))
lpattern(solid) lcolor(%20) lw(medthin) levels(83.4)
coeflabels(1.raceeth = "White" 2.raceeth = "Black" 3.raceeth = "Hispanic" )
byopts(legend(off) col(4) )
ylabel(0 "$0" 300 "$300" 600 "$600" 900 "$900"
1200 "$1200" 1500 "$1500" 1800 "$1800" 2100 "$2100")
yscale(range(0 2100))

ytitle("Annual parental investments per child (2021 real dollars)")
scheme(white_tableau)

;
#delimit cr

esttab M1 M2 M3 M4 using table1.tex, replace frag nogaps booktabs longtable b(a2) se(a2) label ///
order(*raceeth* inca1000) ///
substitute("#" "x" " 0 " "ref." "(.)" "") ///
indicate("Month Fixed Effects = *qintrvmo" "Year Fixed Effects = *qintrvyr") ///
mtitles("Actual" "\shortstack{Controlling\\for family\\characteristics}" ///
"\shortstack{Controlling\\for income}" ///
"\shortstack{Controlling for\\income \& family\\characteristics}")

